

STARSHIP
INSTRUCTIONS

THE OBJECT OF STARSHIP IS TO GET AS HIGH A SCORE AS YOU CAN.
THE POINTS ARE EARNED AS FOLLOWS:
PHASER HIT: 25 POINTS
PHOTON HIT: 50 POINTS

THE ENEMY SHIP LOOKS LIKE TWO BOXES SIDE BY SIDE.

PHASERS ARE FIRED BY PULLING THE TRIGGER. PHOTON TORPEDOES ARE LAUNCHED BY HITTING THE "WORDS" KEY ON THE KEYPAD.

YOU HAVE UNLIMITED PHASER FIRE BUT ONLY HAVE SIX PHOTONS TO START WITH.

A PHOTON WILL DESTROY THE ENEMY ANYWHERE ON THE SCREEN. WHEN YOU HIT THE "WORDS" KEY HOLD IT DOWN UNTIL YOU SEE THE SCREEN BLINKING ON AND OFF. IF YOU DON'T HAVE ANY PHOTON TORPEDOES LEFT YOU WILL HEAR A LOW SOUND THAT IS DIFFERENT THAN THE MOTOR SOUND USUALLY HEARD.

TO FIRE PHASERS PULL THE TRIGGER UNTIL YOU SEE TWO LINES CONVERGE AT THE MIDDLE. THE ENEMY VESSEL CAN ONLY BE DESTROYED BY PHASERS WHEN IT IS ON THE VERTICAL LINE ON THE SCREEN. IF YOU MISS YOU WILL HEAR A HIGH PITCHED SOUND.

PULLING TOWARD YOU ON THE JOYSTICK MAKES THE SHIP GET SMALLER AND PUSHING OUT AWAY FROM YOU MAKES THE SHIP GET BIGGER GIVING THE APPEARANCE OF DISTANCE.

TURNING THE KNOB CLOCKWISE MAKES THE SHIP MOVE TO THE RIGHT; TURNING THE KNOB COUNTER-CLOCKWISE MAKES THE SHIP MOVE TO THE LEFT.

A GOOD SCORE FOR STARSHIP WOULD BE FROM 350-375. A GREAT SCORE WOULD BE FROM ABOUT 425-500.

NOTE: TO MAKE THE GAME LAST LONGER AND TO GET HIGHER SCORES CHANGE LINE 04. MAKE E HIGHER THAN IT IS FOR HIGHER SCORES. MAKE E LOWER FOR LOWER SCORES.

ALSO, EVERY TIME THE ENEMY SHIP FIRES AT YOU, YOU LOSE 2 ENERGY UNITS. THIS MEANS YOU CAN TAKE FOUR HITS FROM HIM AND NOT BE DESTROYED UNTIL YOUR FIFTH HIT. WHEN YOU ARE DESTROYED THE BALLY WILL TELL YOU THAT YOU ARE DESTROYED AND THEN AFTER A PAUSE, IT WILL GIVE YOU YOUR FINAL SCORE.

STARSHIP

```
03 CLEAR;NT=0
04 E=10
05 BC=0;FC=126
06 C=0;B=0
07 P=6;.PHOTON TORP.
10 PRINT "      STARSHIP"
20 PRINT;PRINT"TRIGGER FIRES PHASERS.  TO FIRE A PHOTON TORPEDO HIT 'WORDS'
    KEY ON KEYPAD"
30 PRINT "YOU HAVE 6 PHOTONS TO START"
35 FOR Q=1 TO 2000; NEXT Q
40 CLEAR
41 K=RND(140)-70
42 Y=RND(70)-30
45 LINE 0,-43,4;LINE 0,43,1;LINE 79,0,4,LINE-79,0,1
60 FOR A = 1 TO RND (35)+20
65 BOX RND(160)-80,RND(87)-44,1,1,1
67 NEXT A
70 D=RND(5)+RND(7);S=D-1;.D=DISTANCE
75 &(10)=180
80 &(16)=600;&(18)=600;&(19)=500;&(22)=1;&(23)=100;&(21)=255
81 R=RND(10)
82 IF R=6 GOTO 330
84 IF TR(1)=1 GOTO 170
85 IF JY(1)=1 D=D+1;S=S+1;GOTO 120
90 IF JY(1)=-1 D=D-1;S=S-1;IF D<2D=2;S=D-1
100 IF TR(1)=1 GOTO 170
110 IF &(20)=32 GOTO 300
120 BOX K,Y,D,D,3
125 BOX K+D,Y,S,S,3
126 IF F=1 F=0;BOX K,Y,D,D,3;BOX K+D,Y,S,S,3; GOTO 80
127 F=0
128 F=1
130 IF KN(1)<-50BOX K,Y,D,D,3;BOXK+D,Y,S,S,3;K=K-4;GOTO 120
140 IF KN(1)<50 BOX K,Y,D,D,3;BOX K+D,Y,S,S,3; GOTO 100
150 BOX K,Y,D,D,3;BOX K+D,Y,S,S,3;K=K+4; GOTO 120
160 GOTO 80
170 NT=3
171 FOR Z= 1 TO 10
172 BOX K,Y,D,D,3; BOX K+D,Y,S,S,3
175 &(16)=0;&(17)=0;&(18)=0;&(19)=2; &(21)=15
178 &(20)=15
180 LINE-79,-43,4;LINE 0,0,3;LINE 79,-43,4;LINE 0,0,3
190 NEXT Z
192 LINE-79,-43,4;LINE 0,0,2;LINE 79,-43,4; LINE 0,0,2
197 H=0
200 IF K>4-D IF K<4+DH=1
205 NT=0
210 IF H=0 GOTO 220
215 B=25;GOTO 250
220 BOX K,Y,D,D,3;BOX K+D,Y,S,S,3
221 FOR Z=1 TO 75
225 &(17)=60;&(16)=80;&(21)=15
227 NEXT Z
228 BOX K,Y,D,D,3;BOX K+D,Y,S,S,3
229 F=0;GOTO 120
250 FOR Q=60 TO 131
260 &(21)=15;&(20)=Q
270 NEXT Q
271 FOR Z=1 TO 4+RND(4)
```

```
272 LINE K+RND(6),Y-RND(6)+1,4;LINE K,Y,1
273 MU="V";MU="P"
274 NEXT Z
275 FOR Z=1 TO 50
280 BC=100;BC=100;BC=0
290 NEXT Z
295 &(10)=0;BC=0
296 &(21)=0
297 C=C+B
298 GOSUB 500
299 GOTO 40
300 P=P-1
310 IF P<0P=0;&(21)=255;&(22)=150;&(19)=230;F=0;GOTO 120
315 FOR A=1 TO 10
320 &(10)=0;FOR I=1 TO 50;NEXT I;&(10)=180
322 NEXT A
323 B=50
325 GOTO 250
330 E=E-2
335 U=RND(160)-80
337 BOX K,Y,D,D,3;BOX K+D,Y,S,S,3
340 LINE K,Y,4;LINE U,-43,3
350 FOR Q=129 TO 192
360 &(21)=15;&(20)=Q
380 NEXT Q
390 FOR Z= 1 TO 30
400 BC=98;BC=0;&(10)=0
410 NEXT Z
420 &(20)=180;BC=0
422 LINE K,Y,4;LINE U,-43,3
423 BOX K,Y,D,D,3;BOX K+D,Y,S,S,3
425 IF E>0 F=0;GOTO 120
430 PRINT " YOU WERE DESTROYED"
440 &(21)=0
443 FOR L=1 TO 999; NEXT L
444 CY=40;PRINT"(23 spaces)"
445 GOSUB 500
450 STOP
500 CY=40
505 &(10)=180
510 PRINT"SCORE",
520 PRINT #5,C," PHOTONS ",#1,P
525 FOR L=1 TO 999;NEXT L
530 RETURN
```